

POWER BI — Séance 6

DAX avancé — KPI supply chain et calculs temporels

5 PPA SC — Manager Achats et Supply Chain (MS) · 06/05/2026 · 15h45–17h15


FORMATEUR : CLAUDE-JEAN GIRAULT

GIRAULT-CONCEPTIONS

DURÉE : 1H30

Objectifs de la séance

À l'issue de cette séance, vous serez capables de maîtriser les outils DAX avancés pour piloter la performance supply chain.



The screenshot shows the Microsoft Power BI Desktop interface. The ribbon is set to 'Table tools' > 'Column tools'. The formula bar contains the following DAX formula:

```
1 HighSalesCategory = IF(financials[ Sales] > 100,  
2 "High", "Low")
```

Below the formula bar, a table is displayed with the following columns: 'Month Number', 'Month Name', 'Year', and 'HighSalesCategory'. The table contains 12 rows of data, including months from May 2014, December 2014, November 2013, February 2014, March 2014, October 2014, and September 2014, all with a 'High' sales category.

	Month Number	Month Name	Year	HighSalesCategory
May, 1 May 2014	5	May	2014	High
December 2014	12	December	2014	High
December 2014	12	December	2014	High
November 2013	11	November	2013	High
February 2014	2	February	2014	High
1 March 2014	3	March	2014	High
October 2014	10	October	2014	High
October 2014	10	October	2014	High
September 2014	9	September	2014	High
September 2013	9	September	2013	High
May, 1 May 2014	5	May	2014	High

CALCULATE et filtres

Créer des mesures DAX avancées avec CALCULATE et les fonctions de filtre pour modifier dynamiquement le contexte d'évaluation.

Time Intelligence

Utiliser les fonctions temporelles : cumul annuel YTD, comparaison N-1 et décalages de périodes.

KPI Supply Chain

Calculer et interpréter les indicateurs clés de performance supply chain : taux de service, conformité, annulation et budget.



CALCULATE : Modifier le contexte de filtre

FONCTION FONDAMENTALE DAX

Syntaxe générale

```
CALCULATE(  
  expression,  
  filtre1,  
  filtre2,  
  ...  
)
```

Fonction la plus puissante du DAX. Elle évalue une expression dans un contexte de filtre modifié.

Exemples concrets

```
Achats Livrés =  
CALCULATE(  
  [Total Achats],  
  Achats[Statut_Commande] = "Livrée"  
)
```

```
Achats France =  
CALCULATE(  
  [Total Achats],  
  Fournisseurs[Pays] = "France"  
)
```

```
Achats T1 =  
CALCULATE(  
  [Total Achats],  
  Calendrier[Trimestre] = "T1"  
)
```

💡 Principe clé : CALCULATE remplace ou ajoute des filtres au contexte existant avant d'évaluer l'expression. C'est le cœur de toute mesure conditionnelle en DAX.



Exercice : Mesures de taux – KPI Achats

EXERCICE PRATIQUE

Créer les trois KPI de performance achats suivants dans Power BI :

Taux Livraison à Temps

```
Taux Livraison à Temps =  
DIVIDE(  
    CALCULATE(  
        COUNTROWS(Achats),  
        Achats[Retard_Jours] = 0  
    ),  
    CALCULATE(  
        COUNTROWS(Achats),  
        Achats[Statut_Commande] = "Livrée"  
    )  
)
```

 Objectif : 95%

Taux Conformité

```
Taux Conformité =  
DIVIDE(  
    CALCULATE(  
        COUNTROWS(Achats),  
        Achats[Conformité_Qualité]  
        = "Conforme"  
    ),  
    COUNTROWS(Achats)  
)
```

 Objectif : 98%

Taux Annulation

```
Taux Annulation =  
DIVIDE(  
    CALCULATE(  
        COUNTROWS(Achats),  
        Achats[Statut_Commande]  
        = "Annulée"  
    ),  
    COUNTROWS(Achats)  
)
```

 Objectif : < 3%

 Toujours utiliser `DIVIDE()` plutôt que l'opérateur `/` pour éviter les erreurs de division par zéro.



Time Intelligence : Analyse temporelle

PRÉREQUIS : TABLE CALENDRIER

Les fonctions Time Intelligence nécessitent impérativement une **table Calendrier marquée** dans le modèle de données.

TOTALYTD

```
TOTALYTD(  
  [Total Achats],  
  Calendrier[Date]  
)
```

Cumul des achats depuis le 1er janvier – **Year-To-Date**

SAMEPERIODLASTYEAR


```
CALCULATE(  
  [Total Achats],  
  SAMEPERIODLASTYEAR(  
    Calendrier[Date]  
  )  
)
```

Même période l'**année précédente (N-1)**

DATEADD

```
Achats Mois Précédent =  
CALCULATE(  
  [Total Achats],  
  DATEADD(  
    Calendrier[Date],  
    -1, MONTH  
  )  
)
```

Décaler d'un mois en arrière (M-1)



Exercice : Mesures Time Intelligence

EXERCICE PRATIQUE

Mesures à créer

```
// Cumul annuel
Achats YTD =
TOTALYTD(
    [Total Achats],
    Calendrier[Date]
)

// Ratio vs budget
Achats vs Budget =
DIVIDE(
    [Total Achats],
    SUM(Categories[Budget_Annuel_EUR])
)
```

Variations mensuelles

```
// Écart en valeur
Variation M-1 =
[Total Achats]
- CALCULATE(
    [Total Achats],
    DATEADD(
        Calendrier[Date],
        -1, MONTH
    )
)

// Écart en pourcentage
Variation % =
DIVIDE(
    [Variation M-1],
    CALCULATE(
        [Total Achats],
        DATEADD(
            Calendrier[Date],
            -1, MONTH
        )
    )
)
```

Achats vs Budget
Ratio annuel par catégorie.

Variation M-1
Écart en valeur mensuel.

Achats YTD
Cumul depuis janvier.

Variation %
Écart en pourcentage.



KPI Supply Chain – Tableau de bord

Les indicateurs clés de performance supply chain à implémenter dans Power BI. La table `KPI_Objectifs` sert de référence pour les jauges et comparaisons.

KPI	Formule DAX	Objectif	Priorité
Taux de service	Livraisons à temps / Total livrées	≥ 95%	● Critique
Taux conformité	Conformes / Total	≥ 98%	● Critique
Délai moyen livraison	AVERAGE(Retard_Jours)	< 15 jours	● Important
Taux annulation	Annulées / Total	< 3%	● Important
Achats vs Budget	Total achats / Budget annuel	< 100%	● Important
Part achats durables	Achats ISO / Total	≥ 30%	● Stratégique

95%

Taux de service

Objectif minimum de livraisons à temps

98%

Taux conformité

Objectif qualité fournisseurs

3%

Taux annulation max

Seuil d'alerte à ne pas dépasser

30%

Achats durables




Part minimale d'achats certifiés ISO



Mesures conditionnelles avancées

SWITCH · IF · DISTINCTCOUNT · TOPN

Statut de performance dynamique

```
Statut Performance =  
SWITCH(TRUE(),  
  [Taux Livraison à Temps]  
  >= 0.95,  
  "  Excellent",  
  [Taux Livraison à Temps]  
  >= 0.90,  
  "  Acceptable",  
  "  Critique")
```

Comptage fournisseurs actifs

```
Nb Fournisseurs Actifs =  
DISTINCTCOUNT(  
  Achats[ID_Fournisseur]  
)
```

Meilleur fournisseur

```
Top Fournisseur =  
TOPN(  
  1,  
  ALL(Fournisseurs[Nom_Fournisseur]),  
  [Total Achats],  
  DESC  
)
```

Logique de SWITCH(TRUE())

La syntaxe SWITCH(TRUE(), ...) évalue chaque condition dans l'ordre et retourne la première valeur vraie.

→ Condition 1

≥ 95% →  Excellent

→ Condition 2

≥ 90% →  Acceptable

→ Défaut

Sinon →  Critique

 Préférer SWITCH à des IF imbriqués pour la lisibilité et la performance.



Bonnes pratiques DAX

Respecter ces règles garantit des modèles Power BI maintenables, performants et compréhensibles par toute l'équipe.

1 Nommer les mesures clairement

Éviter « Mesure 1 » ou « Calcul2 ». Utiliser des noms explicites : `Taux_Livraison_Temps`, `Achats_YTD`.

2 Formater immédiatement

Appliquer le format dès la création : %, €, **nombre entier**. Évite les erreurs d'affichage dans les visuels.

3 Commenter les formules complexes

Utiliser `// commentaire` pour expliquer la logique. Indispensable pour la maintenance et le travail en équipe.

4 Tester dans un visuel simple

Valider chaque mesure dans une carte ou un tableau avant de l'intégrer dans un visuel complexe.

5 Utiliser `DIVIDE()` et `CALCULATE`

Toujours `DIVIDE()` au lieu de `/` (division par zéro). Préférer `CALCULATE + filtre` plutôt que `IF` pour les KPI.

6 Regrouper dans la table `_Mesures`

Centraliser toutes les mesures dans une table dédiée `_Mesures` pour une navigation claire dans le modèle.



Synthèse – Séance 6

RÉCAPITULATIF

CALCULATE Modifie le contexte de filtre → mesures conditionnelles puissantes et flexibles	Time Intelligence TOTALYTD, DATEADD, SAMEPERIODLASTYEAR pour l'analyse temporelle
KPI créés Taux livraison · Taux conformité · Taux annulation · Achats YTD · Achats vs Budget	Table KPI_Objectifs Référence pour les jauges et comparaisons – moteur analytique de Power BI



CALCULATE

Modifier le contexte de filtre

Time Intelligence

TOTALYTD, DATEADD, SAMEPERIODLASTYEAR

KPI Supply Chain

Taux service, conformité, annulation, budget

Points clés à retenir

- CALCULATE = cœur de toute mesure conditionnelle DAX
- Les fonctions Time Intelligence nécessitent une table Calendrier marquée
- Toujours utiliser DIVIDE() pour les ratios et taux
- Regrouper les mesures dans la table _Mesures
- La table KPI_Objectifs alimente les jauges de performance

Prochaine séance

27/05 – 09h45

Création de tableaux de bord achats complets dans Power BI

SÉANCE 7

